Using one variable to determine action

Sometimes you have a group of if/else statements that all look at the same variable, like the example below that determines the value of msg depending on the value of num.

```
var num:Number=5;
var msg:String="";
if(num==0) msg="Big Sale";
else if(num==1) msg="One day only";
else if(num==2) msg="Everything on sale";
else if(num==3) msg="Save!";
else msg="Sale";
trace(msg);
```

Start a new Flash ActionScript3 movie and copy the code in the action window. Try different values for num in the first statement. Notice that if the value of num is not 0, 1, 2, or 3 the value of **msg** will be "Sale".

Using switch

The switch statement simpifies writing a group of statements like the one above. The code below does exactly the same as the first example, but is easier to understand and to modify.

```
var num:Number=5;
var msg:String="";
switch(num) {
     case 0: msg="Big Sale"; break;
     case 1: msg="One day only"; break;
     case 2: msg="Everything on sale"; break;
     case 3: msg="Save!"; break;
     default: msg="Sale";
} //switch
trace(msg);
```

The keyword switch is followed by the variable inside parentheses. The body of the switch block is enclosed in curly braces. When a matching value for the variable is found in the case statements, every executable statement is executed until there is a break.

The example below shows another format:

```
var month:int=0;
var msg:String="";
switch (month) {
  case 1: case 2: case 12: msg="Winter"; break;
  case 3: case 4: case 5: msg="Spring"; break;
  case 6: case 7: case 8: msg="Summer"; break;
  case 9: case 10: case 11: msg="Fall"; break;
  default: msg=month+" is not a valid month";
} // switch month
trace(msg);
```

Try this example with different values, including a value that is not a month. Remove one of the breaks and see what happens.

Changing Message

This movie clip shows a message that changes every 40 frames.

- Start a new project.
- Add a text box. Select dynamis text in the property window and name it lblMessage.
- Add the code as shown below:

```
var num:int=0; //which message to show
var delay:int=40; //how long to show each message
var frame:int=0; //counter for how long message has shown
lblMessage.text="Coming soon";
this.addEventListener(Event.ENTER_FRAME,frames);
function frames(e:Event):void {
     frame++; //count frames
     if(frame>delay) nextMessage();
}
function nextMessage():void {
  frame=0; //reset to wait another 40 frames
  var msg:String="";
  num++; //next message
  if(num>2) num=0; //last message, go back to beginning
  switch(num) {
    case 0: msg="Big Sale"; break;
    case 1: msg="One day only"; break;
    case 2: msg="Everything on sale"; break;
  } //switch
  lblMessage.text=msg;
} //nextMessage
```

Which Button?

The switch statement can also be used to determine which button was pressed.

- Start a new project.
- Create a simple Ball movie clip and name the instance ball.
- Add a button called btnLeft and another button called btnRight.
- Add the code as shown below:

```
btnLeft.addEventListener(MouseEvent.CLICK,moveBall);
btnRight.addEventListener(MouseEvent.CLICK,moveBall);
function moveBall(e:MouseEvent):void {
     switch(e.target) {
          case btnLeft: ball.x--; break; //move left
          case btnRight: ball.x++; break; //move right
     } //switch e.target
} //moveBall
```

Run the program and test each button. Add additional buttons to move the ball up, down, or change other properties.